
ДВОЕТАПНА ТРАНСПОРТНА ЗАДАЧА  
ТА ЇЇ AMPL-РЕАЛІЗАЦІЯ

Наведено формулювання двоетапної транспортної задачі закритого типу та обґрунтовано 
умови, за яких система лінійних обмежень задачі є несумісною. Ці умови використано під час роз-
робки AMPLкоду для розв’язання задачі за допомогою сучасного програмного забезпечення для 
задач лінійного програмування. Наведено демонстраційний приклад із результатами розрахунку за 
допомогою програми gurobi NEOSсервера.

Ключові слова: двоетапна транспортна задача, задача лінійного програмування, мова моделю-

Вступ

Формулювання класичної двоетапної транс-
портної задачі [1] передбачає, що вантаж перево-
зиться від постачальників до споживачів тільки 
через проміжні пункти. Схему функціонування 
перевезень вантажу наведено на рис. 1. Проміж-
ними пунктами можуть бути посередницькі 
фірми та різного роду сховища (склади).

Рис. 1. Система «постачальники – проміжні пункти – 
споживачі» у двоетапній транспортній задачі

Для розв’язання двоетапної транспортної за-
дачі застосовують різного роду спеціалізовані 
алгоритми [1; 2]. У статті розглянемо універ-
сальний спосіб розв’язання двоетапної транс-
портної задачі з використанням сучасного про-
грамного забезпечення. Цей спосіб використовує 
опис задачі на мові моделювання AMPL [5] та 
залежить тільки від того, яка з відомих програм 
використовуватиметься для розв’язання задачі 
лінійного програмування. 

Матеріал статті викладено в такому порядку. 
У розділі 1 описано формулювання двоетапної 
транспортної задачі та наведено її властивості. 
У розділі 2 наведено опис задачі на мові моде-
лювання AMPL [5] та результати розрахунків 
для тестового прикладу.

1. Двоетапна транспортна задача  
та її властивості

Нехай у m  пунктах постачання A Am1, ...,   
є a am1, ...,  одиниць продукції, яку потрібно 
перевезти до n  споживачів B Bn1, ..., , задо-
вольнивши їхні потреби b bn1, ..., . Для транс-
портування продукції від постачальників до 
споживачів можна задіяти l  проміжних пунк-
тів D Dl1, ..., . Витрати на перевезення одиниці 
продукції з пункту постачання Ai  до проміж-
ного пункту Dk  позначимо cik , з проміжного 
пункту Dk  до споживача B j  – ckj . Кількість 
продукції, яка перевозиться від постачальника 
Ai  до проміжного пункту Dk , позначимо xik , 

кількість продукції від проміжного пункту до 
споживача – ykj . 

Двоетапна транспортна задача має такий ви-
гляд: знайти

Задача (1)–(5) є задачею лінійного програму-
вання, яка містить m l l n× + ×  змінних xik , ykj  та 
m n l+ +  обмежень загального виду, не враховую-
чи обмежень (5) – умов на невід’ємність усіх змін-
них. Цільова функція (1) задає сумарні витрати на 
транспортування продукції від постачальників до 
споживачів через проміжні пункти. Обмежен-
ня (2) означають транспортування усієї продукції 
a am1, ...,  із пунктів постачання до проміжних 
пунктів, а обмеження (3) – що споживачам потріб-
но доставити необхідну продукцію b bn1, ...,  
з проміжних пунктів. Обмеження (4) задають 
умови на те, щоб уся продукція, яка надходить від 
постачальників до кожного проміжного пункту, 
була обов’язково відправлена споживачам. Це ви-
значає умови на сумісність системи обмежень 
(2)–(5) – системи лінійних рівностей та лінійних 
нерівностей. Звідси випливає таке твердження.

Лема 1. Система обмежень (2)–(5) є несуміс-

ною, якщо a bi j

Доведення. Доведення проведемо методом 
від супротивного. Приймемо, що існують не-
від’ємні x x i m k lik= = ={ }, , , , , ,1 1   та 
y y k l j nkj= = ={ }, , , , , ,1 1  , які задовольня-
ють систему обмежень (2)–(4). Тобто для x ≥ 0  
та y ≥ 0  справджуються такі рівності:

Якщо всі рівності із (6) просумувати за індек-
сом i m=1, , , а всі рівності із (7) просумувати за 
індексом j n=1, , , то отримаємо такі рівності: 

Віднявши від рівності (9) рівність (10) та змі-
нивши порядок індексів у сумах, отримаємо таку 
рівність:

Якщо всі рівності із (8) просумувати за індек-
сом k l=1, , , то отримаємо таку рівність:

0 , що суперечить рівно-

сті (12). Це протиріччя доводить, що вектори  
x  та y  не можуть бути допустимими для обме-
жень (2)–(4), отже, при виконанні умови 

 система обмежень (2)–(5) є несу-

місною. Лему доведено.
Лема 1 дає можливість перевірити вхідні дані 

щодо їх коректності для сформульованої дво-
етапної транспортної задачі. Якщо не виконуєть-
ся умова леми, то тоді виконується умова 

. У цьому випадку система (2)–(5) 

є сумісною і можна переходити до розв’язання 
задачі (1)–(5).

2. AMPL-реалізація задачі (1)–(5)  
та тестовий приклад

AMPL-код для опису двоетапної транспорт-
ної задачі (1)–(5) має такий вигляд:

param m>=2; #Кількість постачальників (пункти A)
param l>=1; #Кількість проміжних пунктів (пункти D)
param n>=2; #Кількість споживачів (пункти B)
#Вартості перевезення одиниці продукції:
param cik{i in 1..m, k in 1..l} >= 0; #від A до D
param ckj{k in 1..l, j in 1..n} >= 0; #від D до B

#Інші дані
param a{i in 1..m} >= 0; #Продукція в A
param b{j in 1..n} >= 0; #Потреби в B
check: sum{i in 1..m} a[i] = sum{j in 1..n} b[j]; # умова леми 1
#Невідомі (продукція, яку потрібно перевезти)
var x{i in 1..m, k in 1..l}  >=0; #від A до D
var y{k in 1..l, j in 1..n}  >=0; #від D до B
minimize f_opt: #Мінімізувати витрати на перевезення продукції:
sum{i in 1..m, k in 1..l} cik[i,k]*x[i,k]+ #від A до D
sum{k in 1..l, j in 1..n} ckj[k,j]*y[k,j]; #від D до B
subject to #за обмежень
con2 {i in 1..m}: #перевезення продукції з A до D
 sum{k in 1..l}x[i,k] = a[i];
con3 {j in 1..n}: #задоволення потреб В з D
 sum{k in 1..l}y[k,j] = b[j];
con4 {k in 1..l}: #всю продукцію потрібно доставити в В
 sum{i in 1..m}x[i,k]-sum{j in 1..n}y[k,j]=0;

Тут оператор param використано для опису 
розмірів та даних задачі; оператор var – для 
опису змінних задачі, де враховано їх невід’єм-
ність; оператор check – для перевірки сумісності 
обмежень (2)–(5), яка визначається лемою 1 та 
вимагає, щоб уся продукція постачальників була 
відправлена споживачам.

Якщо цей AMPL-код доповнити необхідними 
даними для конкретної задачі, то його можна вико-
ристовувати для розв’язання двоетапних транс-
портних задач за допомогою стандартного про-
грамного забезпечення для розв’язання задач 
лінійного програмування. Це можна зробити як за 
допомогою тих програм NEOS-сервера [6] із розді-
лу «Linear Programming», для яких підтримуються 
вхідні формати даних на AMPL, так і за допомогою 
комерційних або з вільним доступом версій AMPL 1.

Тестовий приклад [3]. Виробниче об’єднан-
ня складається з трьох філій: А1, А2, А3, які  
виготовляють однорідну продукцію в обсягах 

відповідно 1000, 1500 та 1200 одиниць на місяць. 
Ця продукція відправляється на два склади  
D1 і D2,  а потім – до п’яти споживачів B1,  B2,  …, 
B5,  попит яких становить відповідно 900, 700, 
1000, 500 і 600 одиниць. Вартості перевезень 
одиниці продукції (в умовних одиницях) від ви-
робників на склади, а потім – зі складів до спо-
живачів наведено в таких таблицях:

Для цього прикладу опис вхідних даних задачі (1)–(5) реалізує AMPL-код: 

data; #Блок вхідних даних, де задаємо:
param m := 3; #Кількість постачальників A
param l := 2; #Кількість проміжних пунктів D
param n := 5; #Кількість споживачів B
#Витрати на перевезення одиниці продукції:
param cik: 1 2 := #від А (↓) до D (→)
         1 2 8
         2 3 5
         3 1 4;
param ckj: 1 2 3 4 5 := #від D (↓) до B (→)
         1 1 3 8 5 4
         2 2 4 5 3 1;

1  Безкоштовні версії програми розміщено за посиланням: http://ampl.com/try-ampl/download-a-free-demo/

param a:= #Продукція в А
1 1000 2 1500 3 1200; #A_1, A_2, A_3
param b:= #Потреби В
1 900 2 700 3 1000 #B_1, B_2, B_3
4 500 5 600; #B_4, B_5
solve; #Розв’язати задачу (1)-(5)
#Роздрукувати значення цільової функції та час розв’язання
display f_opt, _solve_time;
#Роздрукувати значення оптимальних змінних
display x; display y;

Результат розрахунку для тестового прикладу за допомогою відомої програми gurobi на  
NEOS-сервері має такий вигляд:

*************************************************************

   NEOS Server Version 5.0
   Job#     : 6064547
   Password : AonehGLi
   User     : None
   Solver   : lp:Gurobi:AMPL
   Start    : 2018-05-23 00:49:53
   End      : 2018-05-23 00:49:58
   Host     : NEOS HTCondor Pool

   Disclaimer:

   This information is provided without any express or
   implied warranty. In particular, there is no warranty
   of any kind concerning the fitness of this
   information  for any particular purpose.
*************************************************************
File exists
You are using the solver gurobi_ampl.
Checking ampl.mod for gurobi_options...
Executing AMPL.
processing data.
processing commands.
Executing on prod-exec-2.neos-server.org

16 variables, all linear
10 constraints, all linear; 32 nonzeros
 10 equality constraints
1 linear objective; 16 nonzeros.

Gurobi 7.5.1: optimal solution; objective 22100
1 simplex iterations
f_opt = 22100
_solve_time = 0.003998

x :=
1 1   1000
1 2      0
2 1      0
2 2   1500

3 1   1200
3 2      0
;

y :=
1 1   900
1 2   700
1 3   100
1 4   500
1 5     0
2 1     0
2 2     0
2 3   900
2 4     0
2 5   600
;

Gurobi 7.5.1: threads=4
outlev=1
Optimize a model with 10 rows, 16 columns and 32 nonzeros
Coefficient statistics:
  Matrix range     [1e+00, 1e+00]
  Objective range  [1e+00, 8e+00]
  Bounds range     [0e+00, 0e+00]
  RHS range        [5e+02, 2e+03]
Iteration    Objective       Primal Inf.    Dual Inf.      Time
       0    2.2100000e+04   0.000000e+00   0.000000e+00      0s

Solved in 0 iterations and 0.00 seconds
Optimal objective  2.210000000e+04
Gurobi 7.5.1: optimal solution; objective 22100

Розв’язок задачі (1)–(5) для тестового при-
кладу наведено на рис. 2. Для нього значення 
цільової функції дорівнює 22100 умовних 

одиниць, склад D1  завантажений на 2200 оди-
ниць, а склад D2  – на 1500 одиниць.

Якщо умови леми 1 задовольняються, то 
система (2)–(5) є несумісною і програма 
gurobi не переходитиме до розв’язання 
задачі (1)–(5). 

Наприклад, якщо в описі даних оператори 
param a:= #Продукція в А
1 1000 2 1500 3 1200; #A_1, A_2, A_3

замінити на оператори 
param a:= #Продукція в А
1 1000 2 1500 3 1100; #A_1, A_2, A_3

то результат розрахунку на NEOS-сервері має 
такий вигляд:

*************************************************************

   NEOS Server Version 5.0
   Job#     : 6064560
   Password : mMRdIvOn
   User     : None
   Solver   : lp:Gurobi:AMPL
   Start    : 2018-05-23 01:26:09

Рис. 2. Оптимальний план перевезень продукції

   End      : 2018-05-23 01:26:14
   Host     : NEOS HTCondor Pool

   Disclaimer:

   This information is provided without any express or
   implied warranty. In particular, there is no warranty
   of any kind concerning the fitness of this
   information  for any particular purpose.
*************************************************************
check at line 16 of amplin fails:
  check: sum{i in 1 .. m} a[i] == sum{j in 1 .. n} b[j];
xecuting AMPL.
processing data.
processing commands.
Executing on prod-exec-2.neos-server.org
 Error (2) in /opt/ampl/ampl -R amplin

Це означає, що кількість продукції у поста-
чальників на 100 одиниць менша, ніж це потріб-
но споживачам. У цьому випадку система об-
межень є несумісною, про що сигналізує 
AMPL-оператор check.

Висновки

У цій статті описано двоетапну транспортну 
задачу та наведено AMPL-код для її розв’язання 
за допомогою сучасного програмного забезпе-
чення для задач лінійного програмування. Наве-
дено демонстраційний приклад з результатами 
розрахунку. Розроблений AMPL-код використо-
вують під час вивчення навчальної дисципліни 
«Мережеві задачі оптимізації» магістранти 
спеціальностей 8.122 «Комп’ютерні науки» та 

8.121 «Інженерія програмного забезпечення» 
ДВНЗ «Ужгородський національний універси-
тет» [4]. Його планується використати під час 
вивчення навчальної дисципліни «Системи об-
робки економічної інформації» для студентів 
економічного факультету Національного універ-
ситету «Києво-Могилянська академія».

Модифікації двоетапної транспортної зада-
чі (1)–(5) можуть використовувати студенти 
вищих навчальних закладів для розроблення 
власних програмних проектів, пов’язаних із тео-
рією та методами оптимізації. Наприклад, якщо 
обмеження (5) замінити на обмеження, які вра-
ховують нижні та верхні межі на кількість про-
дукції, що перевозиться, то описаний метод не 
зміниться, а спеціалізовані алгоритми потребу-
ватимуть значної корекції.

Petro Stetsyuk, Volodymyr Lyashko, Gabriela Mazyutynets

TWO-STAGE TRANSPORTATION PROBLEM  
AND ITS AMPL-REALIZATION

The paper discusses the two-stage transportation problem of the closed type and investigates its 
properties. The problem is connected with finding an optimal plan for transportation of products from 
suppliers to consumers, using intermediate points. The formulation of the two-stage transportation problem 
is presented in the form of a linear programming problem, and the conditions under which the system of 

linear constraints is incompatible are substantiated. These conditions are connected with checking that  
the total quantity of products from suppliers does not equal the total quantity of products that consumers need. 
It is convenient for using an AMPL check operator to determine when the two-stage transportation problem 
has no solution. 

The AMPL code for solving the two-stage transportation problem with the help of up-to-date software 
for linear programming problems is offered. To describe the AMPL model, the code uses a check operator 
which stops the process of solving the two-stage transportation problem if the linear constraint system is 
incompatible. Otherwise an optimal plan for the transportation of products is printed, which can be found 
with the help of any program for solving the linear programming problem.

A demonstration example with the results of the calculations using the gurobi program from the NEOS 
server is presented. The demonstration example involves the supply of products from three suppliers to five 
consumers, using only two intermediate points. The AMPL code to describe the input data of  
the demonstration example is given, and the output files of the program gurobi for two cases are presented: 
(1) when the check operator allows not to interrupt the process of solving the problem; (2) when the check 
operator reports the incompatibility of the system of linear equations. An optimal plan for transportation of 
products found by the program gurobi for the demonstration example is graphically illustrated.

The developed AMPL code can be used in the study of academic disciplines for graduates of  
the specialities “Computer Science” and “Software Engineering” and students of economic departments. Its 
modifications for the two-stage transportation problem can be used by students of higher educational 
institutions to develop their own software projects related to the theory and methods of optimization.

Keywords: two-stage transportation problem, linear programming problem, AMPL, NEOS server, 
gurobi.

